
3 Syntax von Programmiersprachen

Syntax („Lehre vom Satzbau“)

ñ formale Beschreibung des Aufbaus der „Worte“ und „Sätze“,

die zu einer Sprache gehören;

ñ im Falle einer Programmiersprache Festlegung, wie

Programme aussehen müssen.

Hilfsmittel

Hilfsmittel bei natürlicher Sprache

ñ Wörterbücher;

ñ Rechtschreibregeln, Trennungsregeln, Grammatikregeln;

ñ Ausnahmelisten;

ñ Sprachgefühl.

Hilfsmittel

Hilfsmittel bei Programmiersprachen

ñ Listen von Schlüsselworten wie if, int, else, while...

ñ Regeln, wie einzelne Worte (Tokens) z.B. Namen gebildet

werden.

Frage: Ist x10 ein zulässiger Name für eine Variable (oder

_ab oder A#B oder 0A?B)?...

ñ Grammatikregeln, die angeben, wie größere Komponenten

aus kleineren aufgebaut werden.

Frage: Ist ein while-Statement im else-Teil erlaubt?

ñ Kontextbedingungen.

Beispiel: Eine Variable muss erst deklariert sein, bevor sie

verwendet wird.

Beobachtung

Programmiersprachen sind

ñ formalisierter als natürliche Sprache

ñ besser für maschinelle Verarbeitung geeignet.

Syntax vs. Semantik

Semantik („Lehre von der Bedeutung“)

ñ Ein Satz einer (natürlichen) Sprache verfügt zusätzlich über

eine Bedeutung, d.h teilt einem Hörer/Leser einen

Sachverhalt mit ( Information)

ñ Ein Satz einer Programmiersprache, d.h. ein Programm

verfügt ebenfalls über eine Bedeutung...

Syntax vs. Semantik

Die Bedeutung eines Programms ist

ñ alle möglichen Ausführungen der beschriebenen

Berechnung ( operationelle Semantik); oder

ñ die definierte Abbildung der Eingaben auf die Ausgaben

( denotationelle Semantik).

Achtung!

Ist ein Programm syntaktisch korrekt, heißt das noch lange nicht,

dass es auch das „richtige“ tut, d.h. semantisch korrekt ist!!!

3.1 Reservierte Wörter

ñ int

=⇒ Bezeichner für Basistypen;

ñ if, else, then, while. . .

=⇒ Schlüsselwörter für Programmkonstrukte;

ñ (,), ",’, {,}, ,,;

=⇒ Sonderzeichen;

3.2 Was ist ein erlaubter Name?

Schritt 1:

Festlegung erlaubter Zeichen:

letter ::= $ | _ | a | . . . | z | A | . . . | Z
digit ::= 0 | . . . | 9

ñ letter und digit bezeichnen Zeichenklassen, d.h. Mengen

von Zeichen, die gleich behandelt werden.

ñ Das Symbol „|“ trennt zulässige Alternativen.

ñ Das Symbol „. . .“ repräsentiert die Faulheit, alle Alternativen

wirklich aufzuzählen.

3.2 Was ist ein erlaubter Name?

Schritt 2:

Festlegung der Zeichenanordnung:

name ::= letter ( letter | digit )*

ñ Erst kommt ein Zeichen der Klasse letter, dann eine

(eventuell auch leere) Folge von Zeichen entweder aus letter

oder aus digit.

ñ Der Operator „*“ bedeutet „beliebig oft wiederholen“

(„weglassen“ ist 0-malige Wiederholung).

ñ Der Operator „*“ ist ein Postfix-Operator. Das heißt, er steht

hinter seinem Argument.

Wir definieren hier MiniJava. Eigentli-
ches Java erlaubt mehr Namen (z.B. sind
UTF8-Symbole erlaubt).

Beispiele

_178

Das_ist_kein_Name

x

_

$Password$

...sind legale Namen.

5ABC

!Hallo!

x’

a=b

-178

...sind keine legalen Namen.

Achtung

Reservierte Wörter sind als Namen verboten.

3.3 Ganze Zahlen

Werte, die direkt im Programm stehen, heißen Konstanten.

Ganze nichtnegative Zahlen bestehen aus einer nichtleeren Folge

von Ziffern:

number ::= digit digit*

ñ Wie sähe die Regel aus, wenn wir führende Nullen verbieten

wollen?

Beispiele

17

12490

42

0

00070

...sind int-Konstanten

"Hello World!"

0.5e+128

...sind keine int-Konstanten

Reguläre Ausdrücke

Ausdrücke, die aus Zeichen(-klassen) mithilfe von

| (Alternative)

* (Iteration)

(Konkatenation) sowie

? (Option)

...aufgebaut sind, heißen reguläre Ausdrücke

( Automatentheorie).

Der Postfix-Operator „?“ besagt, dass das Argument eventuell

auch fehlen darf, d.h. einmal oder keinmal vorkommt.

Gelegentlich sind auch ε, d.h. das „leere Wort“
sowie �, d.h. die leere Menge zugelassen.

Die Alternative hat eine geringere Bindungsstär-
ke als die Konkatenation. D.h. ab|c steht für die
Wörter ab oder c und nicht für ab oder ac.

Beispiele
Reguläre Ausdrücke reichen zur Beschreibung einfacher Mengen

von Worten aus.

ñ ( letter letter )*

=⇒ alle Wörter gerader Länge (über

$,_,a,...,z,A,...,Z);

ñ letter* test letter*

=⇒ alle Wörter, die das Teilwort test enthalten;

ñ _ digit* 17

=⇒ alle Wörter, die mit _ anfangen, dann eine beliebige

Folge von Ziffern aufweisen, die mit 17 aufhört;

ñ exp ::= (e|E)(+|-)? digit digit*

float ::= digit digit* exp |
digit* ( digit . | . digit) digit* exp?

=⇒ alle Gleitkommazahlen...

Programmverarbeitung

1. Phase ( Scanner)

Identifizierung von

ñ reservierten Wörtern,

ñ Namen,

ñ Konstanten

Ignorierung von

ñ Whitespace,

ñ Kommentaren

=⇒ Input wird mit regulären Ausdrücken verglichen und dabei in

Wörter („Tokens“) zerlegt.

2. Phase ( Parser)

Analyse der Struktur des Programms.

3.4 Struktur von Programmen

Programme sind hierarchisch aus Komponenten aufgebaut. Für

jede Komponente geben wir Regeln an, wie sie aus anderen

Komponenten zusammengesetzt sein können.

program ::= decl* stmt*

decl ::= type name ( , name )* ;

type ::= int

ñ Ein Programm besteht aus einer Folge von Deklarationen,

gefolgt von einer Folge von Statements.

ñ Eine Deklaration gibt den Typ an, hier: int, gefolgt von

einer Komma-separierten Liste von Variablennamen.

Anweisungen

stmt ::= ; | { stmt* } |
name = expr; | name = read(); |
write( expr ); |
if ( cond ) stmt |
if ( cond ) stmt else stmt |
while ( cond ) stmt

ñ Ein Statement ist entweder „leer“ (d.h. gleich ;) oder eine

geklammerte Folge von Statements;

ñ oder eine Zuweisung, eine Lese- oder Schreiboperation;

ñ eine (einseitige oder zweiseitige) bedingte Verzweigung;

ñ oder eine Schleife.

Ausdrücke

expr ::= number | name | ( expr ) |
unop expr | expr binop expr

unop ::= -

binop ::= - | + | * | / | %

ñ Ein Ausdruck ist eine Konstante, eine Variable oder ein

geklammerter Ausdruck

ñ oder ein unärer Operator, angewandt auf einen Ausdruck,

ñ oder ein binärer Operator, angewandt auf zwei

Argumentausdrücke.

ñ Einziger unärer Operator ist (bisher) die Negation.

ñ Mögliche binäre Operatoren sind Addition, Subtraktion,

Multiplikation, (ganzzahlige) Division und Modulo.

Bedingungen

cond ::= true | false | ( cond ) |
expr comp expr |
bunop ( cond ) | cond bbinop cond

comp ::= == | != | <= | < | >= | >
bunop ::= !

bbinop ::= && | ||

ñ Bedingungen unterscheiden sich dadurch von Ausdrücken,

dass ihr Wert nicht vom Typ int ist sondern true oder

false (ein Wahrheitswert – vom Typ boolean).

ñ Bedingungen sind darum Konstanten, Vergleiche

ñ oder logische Verknüpfungen anderer Bedingungen.

Beispiel

int x;
x = read();
if (x > 0)

write(1);
else

write(0);

Die hierarchische Untergliederung von Programm-Bestandteilen

veranschaulichen wir durch Syntax-Bäume.

Syntaxbäume

Syntaxbäume für x > 0 sowie write(0); und write(1);

cond

expr

name comp

expr

number

>x 0

stmt

expr

num

0(write ) ;

stmt

expr

num

1(write ) ;

Blätter: Wörter/Tokens

innere Knoten: Namen von Programmbestandteilen

Beispiel

Der komplette Syntaxbaum unseres Beispiels:

program

decl

type name

stmt

name

stmt

cond

expr

name comp

expr

num

stmt

expr

num

stmt

expr

num

int x ; x = read ( ) ; if ( x > 0 ) write ( 1 ) ; else write ( 0 ) ;

Bemerkungen

ñ Die vorgestellte Methode der Beschreibung von Syntax heißt

EBNF-Notation (Extended Backus Naur Form Notation).

ñ Ein anderer Name dafür ist erweiterte kontextfreie

Grammatik ( Linguistik, Automatentheorie).

ñ Linke Seiten von Regeln heißen auch Nichtterminale.

ñ Tokens heißen auch Terminale.

Noam Chomsky,
MIT

John Backus, IBM
Turing Award

(Erfinder von Fortran)

Peter Naur,
Turing Award

(Erfinder von Algol60)

Kontextfreie Grammatiken

Achtung:

ñ Die regulären Ausdrücke auf den rechten Regelseiten

können sowohl Terminale wie Nichtterminale enthalten.

ñ Deshalb sind kontextfreie Grammatiken mächtiger als

reguläre Ausdrücke.

Beispiel:

L = {ε,ab,aabb,aaabbb, . . .}

lässt sich mithilfe einer Grammatik beschreiben:

A ::= ( a A b )?

Kontextfreie Grammatiken

Syntaxbaum für das Wort aaabbb:

A

A

A

A

aaa b b b

Für L gibt es aber keinen regulären Ausdruck

( Automatentheorie).

